
1 Getting set up

For this lab, we will again be using Sketchpad, a site for building processing
sketches online using processing.js.

1. Go to http://cisc1600.sketchpad.cc in your browser

2. Log in to the account that you created last week. If your login is not working,
you might have created your account last week at http://sketchpad.cc.
If so, create a new account at http://cisc1600.sketchpad.cc.

1.1 Important information (reminder)!

For this lab, you will be creating a different “sketch” for each section (except this
first one). Each time you press the “Render” button (the button with the “play”
symbol on it), Sketchpad saves a version of your sketch and makes it accessible
via a URL. Please make sure that you do this at least once after completing
each step in the lab so that your work is saved properly. At the end of the lab,
I will ask you to share the URLs of all of your sketches with me and I will go
through the saved points.

1.2 Create a project in Thimble with links to your sketches

Create a new project in Thimble for this lab. In the HTML file, there should be
a list for each sketch and a nested sub-list containing the individual steps for
that sketch. After completing and rendering each step, copy the link to that
step into your thimble project. For details, see the previous lab.

2 Interactivity: mouse variables

In this sketch, we will use the mouse variables that Processing automatically
populates for us. We have already seen mouseX and mouseY in the last lab, this
step we will use mousePressed and mouseButton.

1. Create a new sketch. Change the name to “Lab 5, step 2”. Replace the
default code with the following:

// Lab 5, Step 2.1: Create a new sketch
void setup() {

size(300, 300);
}

void draw() {
background(204);
rect(width/4, height/4, width/2, height/2);

}

2. Change the color of the square if the mouse is clicked

• Insert an if statement in the draw() function between the two existing
lines. The call to rect() should be outside of the if statement’s
braces ({})

• The if statement should test whether the variable mousePressed has
the value of true

• If that condition is met, the current fill color should be set to gray
level 230 using the fill() function

• Otherwise (else), the fill color should be set to gray level 255 using
the fill() function

• Update the comment at the top of the sketch and “Render” it

3. Change the color of the square only if the mouse is clicked inside of it

• We will modify the code from the previous step to only change the
color of the box if the mouse is clicked inside of it

• There are four conditions that must be met when the mouse is inside
of the box, one for each edge of the box. Figure out what they are.

• In order to figure out whether all four conditions are met at the same
time, AND them all together using three instances of the && operator

• Using a fourth && operator, combine these four conditions with the
original condition in the if statement that tests if the mouse button
is pressed

• Try clicking inside the box and above, below, left, and right of it.
Make sure that the box only changes color when you click inside of it.

• Update the comment at the top of the sketch and “Render” it

4. Change the color of the square depending on which mouse button is clicked

• Now we are going to add an additional if statement inside of your
existing if statement to change the fill color of the box based on
which mouse button is pressed

• Processing defines three variables representing mouse buttons: LEFT,
RIGHT, and CENTER. To test which mouse button is clicked, use a test
such as mouseButton == LEFT

• Add this new if statement so that it is run if the mouse is clicked
inside of the box

• Write the if statement so that if the left mouse button is clicked,
the code sets the fill color of the rectangle to 230 using the fill()
function

• Otherwise (else), set the fill color to 170 using the fill() function

• Try clicking different mouse buttons and make sure that the color
changes appropriately

• Update the comment at the top of the sketch and “Render” it

3 Interactivity: keyboard variables

Processing also populates several keyboard variables for you, depending on what
the user is typing: keyPressed, key, and keyCode. We will also use the text()
function to write on the canvas.

1. Create a new sketch. Change the name to “Lab 5, step 3”. Replace the
default code with the following:

// Lab 5, Step 3.1: Create a new sketch
void setup() {

size(300, 300);
textSize(60);
textAlign(CENTER);

}

void draw() {
background(204);
text("None", width/2, height/2);

}

2. Write a different message to the canvas when a key is pressed

• Write an if statement in draw() after the call to background(). The
condition should test whether the variable keyPressed is true. To
test for equality, use the == operator.

• Move the existing call to text() inside the else clause of the if
statement, so that it is only called if a key is NOT pressed.

• If a key IS pressed, add a new call to text() that displays the string
”Pressed”

• Try out the sketch. Make sure that it says “Pressed” when you
type letter keys. What does it say when you press Shift? Control?
Backspace? Enter? Up? Down?

• Update the comment at the top of the sketch and “Render” it

3. Show which key is pressed

• Update your code so that instead of always printing the same message,
it prints the key that was pressed.

• Use the variable key in the call to text() when the if condition is
true

• Try out the sketch. Make sure that it prints the right character when
you type letter keys. What does it say when you press Shift? Control?
Backspace? Enter?

• Update the comment at the top of the sketch and “Render” it

4. Only show printable keys

• Some keys on the keyboard don’t correspond to characters that we
would want to print, for example, Shift, Backspace, Up, Down. At the
moment, when one of those keys is pressed, no character is printed
(actually a blank character is printed)

• Update the test in your if statement so that it tests that a key is
pressed AND that the variable key is not equal to the value CODED.
If it is equal to CODED, then they key is not printable. To test for
inequality, use the != operator.

• Try out the sketch. Make sure that it prints the right character when
you hold letter keys and that it prints “None” when you press Shift,
Control, Up, and Right, but not for Backspace, and Enter (they are
printable). What happens when you hold Shift and press the “1” key
to type an exclamation point?

• Update the comment at the top of the sketch and “Render” it

5. Show the last key that was pressed

• Now we will use a variable to save the value of the last key that was
pressed

• Insert the following line BEFORE the setup() function:

char lastKey = ’ ’;

This declares a new variable called lastKey that holds a single char-
acter. It also stores the space character in the variable to begin with.
We will read and write it inside the draw() function, but by declaring
it outside of any functions, it becomes a global variable, meaning that
its value will be saved between calls to draw()

• Now when the if condition is true, instead of calling text() directly,
save the value of key in the new lastKey variable that you just
created

• Remove the else clause entirely

• Write a new call to text() after the end of the if statement drawing
the character lastKey

• Try out the sketch. Make sure that it prints the right character when
you type letter keys, and that they remain on the screen when you
let go of the key.

• Update the comment at the top of the sketch and “Render” it

4 Interactivity: mouse events

In this sketch, we will define new functions to handle various mouse events:
mousePressed(), mouseReleased(), mouseMoved(), and mouseDragged(). If
we create a function with one of these standard names, it will be called when a
relevant mouse event happens.

1. Create a new sketch. Change the name to “Lab 5, step 4”. Replace the
default code with the following:

// Lab 5, Step 4.1: Create a new sketch
int gray = 0;

void setup() {
size(300, 300);

}

void draw() {
background(gray);

}

void mousePressed() {
}
void mouseReleased() {
}

2. Lighten the background color each time the mouse is pressed

• Inside the mousePressed() function, set the value of the gray variable
to be its current value plus 20.

• Test the sketch. What happens when you click the mouse on the
canvas? What happens when you click and hold?

• Update the comment at the top of the sketch and “Render” it

3. Lighten the background color each time the mouse is released

• Delete the mousePressed() function you just created

• Inside the mouseReleased() function, set the value of the gray vari-
able to be its current value plus 20.

• Test the sketch. What happens when you click the mouse on the
canvas? What happens when you click and hold?

• Update the comment at the top of the sketch and “Render” it

4. Draw a square at each place the mouse is clicked

• Update the sketch to have the following contents only

// Lab 5, Step 4.4: Draw a square at each place the mouse is clicked
void setup() {

size(300, 300);
fill(0, 102);

}

void draw() {
} // Empty draw() keeps the program running

void mousePressed() {
}

• Inside the mousePressed() function, call rect() to draw a 30× 30
square at the current mouse coordinates (remember the variables
mouseX and mouseY from the previous lab).

• Test the sketch. What happens when you click on the canvas? What
happens when you click close to an existing square? Why is that?

5. Optional challenge: Draw a square on the canvas that the user can drag
to a new position.

• Clear your existing code and create appropriate setup(), draw() and
mouseDragged() functions.

• Create two global variables to store the current coordinates of the
square

• Update the coordinates inside the mouseDragged() function, but only
if the click is inside of the square

• Remember that the user probably won’t click right on the point
specified by the coordinates that you store for the square. You’ll need
to figure out where in the square they have clicked and compensate
for the offset from that point before updating the coordinates.

5 Interactivity: keyboard events

The keyboard events keyPressed() and keyReleased() act similarly to the cor-
responding mouse events. Let’s use them to make a little interactive “game”.

1. Create a new sketch. Change the name to “Lab 5, step 5”. Replace the
default code with the following:

// Lab 5, Step 5.1: Create a new sketch
int squareSize = 30;
int squareX, squareY;

void setup() {
size(300, 300);
rectMode(CENTER);
squareX = width/2;
squareY = height/2;

}

void draw() {
background(204);
rect(squareX, squareY, squareSize, squareSize);

}

void keyPressed() {
}

2. Make a square grow each time you hit the enter key

• Add an if statement inside the keyPressed() function. Its condition
should be that the variable key is equal to the value ENTER

• If the condition is met, increase the value of the variable squareSize
by 10

• Test the sketch. The square should grow when you press the “Enter”
key, but not any other key

• Update the comment at the top of the sketch and “Render” it

3. Move a square around the screen using the arrow keys

• Now insert a second if statement after the end of the first if state-
ment. Its condition should be that the variable key is equal to the
value CODED and that the variable keyCode is equal to the value UP.
In this case, you should move the square up by changing the value
of either squareX or squareY by 10. Figure out which one to change
and whether to increase it or decrease it.

• Add an else if clause that tests whether key is CODED and keyCode
is DOWN. Figure out which variable to change and which direction to
change it to move the square down.

• Add similar else if clauses for LEFT and RIGHT

• Update the comment at the top of the sketch and “Render” it

4. Optional challenge: Draw a different shape based on the last key pressed

• Add a variable called lastKey of type char

• In keyPressed() add code at the end so that if key is not equal to
CODED, save it to lastKey

• Update the draw() function to draw a different shape depending on
the value of lastKey. If it is ’c’ draw a circle, ’e’ draw an ellipse, ’l’
draw a line, ’p’ draw a point, ’t’ draw a triangle. Otherwise, draw a
square.

• Update the comment at the top of the sketch and “Render” it

6 Animation: vector graphics

Vector graphics are created using a set of drawing instructions. This allows
them to look crisp and smooth after transformations like growing, rotating, and
stretching. They are mainly only practical for drawings, however, as photos are
not readily transformed into vector representations. In this example, we will
animate a helicopter.

1. Create a new sketch. Change the name to “Lab 5, step 6”. Replace the
default code with the following:

// Lab 5, Step 6.1: Create a new sketch
void setup() {

size(500, 500);
frameRate(20);
fill(128, 128, 255);

}

void draw() {
background(255);
drawHelicopter(mouseX, mouseY);

}

void drawHelicopter(int x, int y) {
// Draw body as quadrangle
strokeWeight(2); // The thickness of the lines
quad(-40 + x, -40 + y, 100 + x, 0 + y,

100 + x, -20 + y, -100 + x, 20 + y);

// Draw the main rotor as a line
strokeWeight(4); // The thickness of the lines
line(-120 + x, -20 + y, -40 + x, -40 + y);
line(-40 + x, -40 + y, 40 + x, -60 + y);

// Draw the tail rotor as a line
strokeWeight(2.5); // The thickness of the lines
line(120 + x, -20 + y, 100 + x, -20 + y);
line(100 + x, -20 + y, 80 + x, -20 + y);

}

• Run the sketch. Move the mouse around the canvas.

2. Animate the rotors

• Notice that each rotor is drawn using two separate lines. Comment out
one call to line() at a time to see which part each one corresponds
to

• Create a global variable called frame that keeps track of the number
of frames that have been drawn

• Add a line to the end of the draw() function to add 1 to frame each
time draw() is called

• Add an if statement before each call to line(), so that line()
is only called if the variable frame has certain values. Figure out
what values frame should have so that the rotor appears to alternate
between the two lines. Hint: use the mod operator %

• Fly the helicopter around

• Update the comment at the top of the sketch and “Render” it

3. Optional challenge: Change the scale of the helicopter

• Create a global variable called scale that is initially set to 1

• In the calls to quad, line, and strokeWeight, wherever there is a
hard-coded number, multiply that number by scale

• Render the sketch and make sure that the helicopter still looks the
same

• Change the value of scale to 2 and see what happens

• Change it to 0.5 and see what happens

• Add a keyPressed() function so that when the key ’-’ is pressed,
scale is decreased and when ’+’ OR ’=’ is pressed, scale is increased.
Implement OR using the || operator

• Update the comment at the top of the sketch and “Render” it

4. Optional challenge: Control the helicopter with the arrow keys

• Give the helicopter a velocity in the x direction and a separate velocity
in the y direction

• Make a keyPressed() listener that translates arrow key presses into
changes in the appropriate velocity

• The helicopter should change position in x and y by a distance equal
to its velocity each frame.

7 Animation: bitmap graphics

Bitmap graphics can be more realistic than vector graphics, but are less flexible.
We will use them here to make a non-realistic animation of our walking stick
figure from the JavaScript lab.

1. Create a new sketch. Change the name to “Lab 5, step 7”. Replace the
default code with the following:

// Lab 5, Step 7.1: Create a new sketch
int frame = 0;
PImage[] frames = new PImage[4];

void setup() {
size(300, 300);
frameRate(10);

}

10



void draw() {
background(255);

}

2. Add animation frames to the sketch

• Download http://m.mr-pc.org/teaching/cisc1600/walkRight.zip
and unzip the files to a directory

• Click on the “files” link in the top right corner of the page on Sketchpad
and upload the four gif files to the site

• As described in the file window where you uploaded your files to, add
a single @pjs call at the top of your file to preload all of your gifs.
The format of the call will be something like this

/* @pjs preload=”/static/uploaded˙resources/p.20231/fr0.gif,/static/...”; */

but the directory of your images will include a different number than
this one. There should just be one set of quotation marks (opening
and closing) and the file locations should be separated by commas.
This makes sure that the files get loaded by the browser before your
sketch runs.

• Follow the other instruction in the file window to load the images
into the frames array inside the setup() function. The code should
look something like this, but again with different numbers

frames[0] = loadImage("/static/uploaded_resources/p.20231/fr0.gif");
frames[1] = loadImage("/static/uploaded_resources/p.20231/fr1.gif");
frames[2] = loadImage("/static/uploaded_resources/p.20231/fr2.gif");
frames[3] = loadImage("/static/uploaded_resources/p.20231/fr3.gif");

3. Draw the current frame and update the frame counter.

• Draw the current frame at position (50, 50) using the image() func-
tion, which takes three arguments: image, x, y. The image that you
pass in should be the frame-th element of the frames array, which
you can access with the syntax frames[frame].

• Add 1 to the frame variable and if it becomes larger than frames.length,
set it to 0. You can use the mod (%) operator for this or an if state-
ment.

4. Make the stick figure walk across the canvas

• Create global variables to store the x and y coordinates of the stick
figure. Don’t forget to initialize them to reasonable values

• Increment one or both of them every frame

• Draw the stick figure at those coordinates using the image() function

5. Optional challenge: Make the stick figure wrap around to the other side
of the scree when it walks off one side

6. Optional challenge: Make the stick figure turn around and walk back
when it gets to the edge of the screen. Download the left-facing frames
from http://m.mr-pc.org/teaching/cisc1600/walkLeft.zip . You’ll
have to increase the size of the frames array

7. Optional challenge: Make the stick figure turn right and take a step to
the right when you press the right arrow. Make it turn to the left and take
a step to the left when you press the left arrow.

